

Multinomial Logit Models

Ying So and Warren F. Kuhfeld
SAS Institute Inc., Cary, NC

ABSTRACT

Multinomial logit models are used to model relationships between a polytomous response variable and a set of regressor variables. The term "multinomial logit model" includes, in a broad sense, a variety of models. The cumulative logit model is used when the response of an individual unit is restricted to one of a finite number of ordinal values. Generalized logit and conditional logit models are used to model consumer choices. This article focuses on the statistical techniques for analyzing discrete choice data and discusses fitting these models using SAS/STAT® software.

Introduction

Multinomial logit models are used to model relationships between a polytomous response variable and a set of regressor variables. These polytomous response models can be classified into two distinct types, depending on whether the response variable has an ordered or unordered structure.

In an ordered model, the response Y of an individual unit is restricted to one of m ordered values. For example, the severity of a medical condition may be: none, mild, and severe. The cumulative logit model assumes that the ordinal nature of the observed response is due to methodological limitations in collecting the data that results in lumping together values of an otherwise continuous response variable (McKelvey and Zavoina 1975). Suppose Y takes values y_1, y_2, \dots, y_m on some scale, where $y_1 < y_2 < \dots < y_m$. It is assumed that the observable variable is a categorized version of a continuous latent variable U such that

$$Y = y_i \Leftrightarrow \alpha_{i-1} < U \leq \alpha_i, i = 1, \dots, m$$

where $-\infty = \alpha_0 < \alpha_1 < \dots < \alpha_m = \infty$. It is further assumed that the latent variable U is determined by the explanatory variable vector \mathbf{x} in the linear form $U = -\beta'\mathbf{x} + \epsilon$, where β is a vector of regression coefficients and ϵ is a random variable with a distribution function F . It follows that

$$\Pr\{Y \leq y_i | \mathbf{x}\} = F(\alpha_i + \beta'\mathbf{x})$$

If F is the logistic distribution function, the cumulative model is also known as the proportional odds model. You can use PROC LOGISTIC or PROC PROBIT directly to fit the cumulative logit models. Although the cumulative model is the most widely used model for ordinal response data, other useful models include the adjacent-categories logit model and the continuation-ratio model (Agresti 1990).

In an unordered model, the polytomous response variable does not have an ordered structure. Two classes of models, the generalized logit models and the conditional logit models, can be used with nominal response data. The generalized logit model consists of a combination of several binary logits estimated simultaneously. For example, the response variable of interest is the occurrence or nonoccurrence of infection after a Caesarean section with two types of (I,II) infection. Two binary logits are considered: one for type I infection versus no infection and the other for type II infection versus no infection. The conditional logit model has been used in biomedical research to estimate relative risks in matched case-control studies. The nuisance parameters that correspond to the matched sets in an unconditional analysis are eliminated by using a conditional likelihood that contains only the relative risk parameters (Breslow and Days 1980). The conditional logit model was also introduced by McFadden (1973) in the context of econometrics.

In studying consumer behavior, an individual is presented with a set of alternatives and asked to choose the most preferred alternative. Both the generalized logit and conditional logit models are used in the analysis of discrete choice data. In a conditional logit model, a choice among alternatives is treated as a function of the characteristics of the alternatives, whereas in a generalized logit model, the choice is a function of the characteristics of the individual making the choice. In many situations, a mixed model that includes both the characteristics of the alternatives and the individual is needed for investigating consumer choice.

Consider an example of travel demand. People are asked to choose between travel by auto, plane or

public transit (bus or train). The following SAS® statements create the data set TRAVEL. The variables AUTOTIME, PLANTIME, and TRANTIME represent the total travel time required to get to a destination by using auto, plane, or transit, respectively. The variable AGE represents the age of the individual being surveyed, and the variable CHOSEN contains the individual's choice of travel mode.

```
data travel;
  input autotime plantime trantime age chosen $;
  cards;
10.0      4.5      10.5      32  Plane
 5.5      4.0      7.5      13  Auto
 4.5      6.0      5.5      41  Transit
 3.5      2.0      5.0      41  Transit
 1.5      4.5      4.0      47  Auto
10.5      3.0      10.5      24  Plane
 7.0      3.0      9.0      27  Auto
 9.0      3.5      9.0      21  Plane
 4.0      5.0      5.5      23  Auto
22.0      4.5      22.5      30  Plane
 7.5      5.5      10.0      58  Plane
11.5      3.5      11.5      36  Transit
 3.5      4.5      4.5      43  Auto
12.0      3.0      11.0      33  Plane
18.0      5.5      20.0      30  Plane
23.0      5.5      21.5      28  Plane
 4.0      3.0      4.5      44  Plane
 5.0      2.5      7.0      37  Transit
 3.5      2.0      7.0      45  Auto
12.5      3.5      15.5      35  Plane
 1.5      4.0      2.0      22  Auto
;
```

In this example, AUTOTIME, PLANTIME, and TRANTIME are alternative-specific variables, whereas AGE is a characteristic of the individual. You use a generalized logit model to investigate the relationship between the choice of transportation and AGE, and you use a conditional logit model to investigate how travel time affects the choice. To study how the choice depends on both the travel time and age of the individual, you need to use a mixed model that incorporates both types of variables.

A survey of the literature reveals a confusion in the terminology for the nominal response models. The term "multinomial logit model" is often used to describe the generalized logit model. The mixed logit is sometimes referred to as the multinomial logit model in which the generalized logit and the conditional logit models are special cases.

The following sections describe discrete choice models, illustrate how to use SAS/STAT software to fit these models, and discuss cross-alternative effects.

Modeling Discrete Choice Data

Consider an individual choosing among m alternatives in a choice set. Let Π_{jk} denote the probability that

individual j chooses alternative k , let X_j represent the characteristics of individual j , and let Z_{jk} be the characteristics of the k th alternative for individual j . For example, X_j may be an age and each Z_{jk} a travel time.

The generalized logit model focuses on the individual as the unit of analysis and uses individual characteristics as explanatory variables. The explanatory variables, being characteristics of an individual, are constant over the alternatives. For example, for each of the m travel modes, $X_j = (1 \text{ age})'$, and for the first subject, $X_1 = (1 \ 32)'$. The probability that individual j chooses alternative k is

$$\Pi_{jk} = \frac{\exp(\beta'_k X_j)}{\sum_{i=1}^m \exp(\beta'_i X_j)} = \frac{1}{\sum_{i=1}^m \exp[(\beta_i - \beta_k)' X_j]}$$

β_1, \dots, β_m are m vectors of unknown regression parameters (each of which is different, even though X_j is constant across alternatives). Since $\sum_{k=1}^m \Pi_{jk} = 1$, the m sets of parameters are not unique. By setting the last set of coefficients to null (that is, $\beta_m = 0$), the coefficients β_k represent the effects of the X variables on the probability of choosing the k th alternative over the last alternative. In fitting such a model, you estimate $m - 1$ sets of regression coefficients.

In the conditional logit model, the explanatory variables Z assume different values for each alternative and the impact of a unit of Z is assumed to be constant across alternatives. For example, for each of the m travel modes, $Z_{jk} = (\text{time})'$, and for the first subject, $Z_{11} = (10)'$, $Z_{12} = (4.5)'$, and $Z_{13} = (10.5)'$. The probability that the individual j chooses alternative k is

$$\Pi_{jk} = \frac{\exp(\theta' Z_{jk})}{\sum_{i=1}^m \exp(\theta' Z_{ji})} = \frac{1}{\sum_{i=1}^m \exp[\theta'(Z_{ji} - Z_{jk})]}$$

θ is a single vector of regression coefficients. The impact of a variable on the choice probabilities derives from the difference of its values across the alternatives.

For the mixed logit model that includes both characteristics of the individual and the alternatives, the choice probabilities are

$$\Pi_{jk} = \frac{\exp(\beta'_k X_j + \theta' Z_{jk})}{\sum_{i=1}^m \exp(\beta'_i X_j + \theta' Z_{ji})}$$

$\beta_1, \dots, \beta_{m-1}$ and $\beta_m \equiv 0$ are the alternative-specific coefficients, and θ is the set of global coefficients.

Fitting Discrete Choice Models

The CATMOD procedure in SAS/STAT software directly fits the generalized logit model. SAS/STAT software does not yet have a procedure that is specially

designed to fit the conditional or mixed logit models. However, with some preliminary data processing, you can use the PHREG procedure to fit these models.

The PHREG procedure fits the Cox proportional hazards model to survival data (refer to SAS Technical Report P-229). The partial likelihood of Breslow has the same form as the likelihood in a conditional logit model.

Let z_l denote the vector of explanatory variables for individual l . Let $t_1 < t_2 < \dots < t_k$ denote k distinct ordered events times. Let d_i denote the number of failures at t_i . Let s_i be the sum of the vectors z_l for those individuals that fail at t_i , and let \mathcal{R}_i denote the set of indices for those who are at risk just before t_i .

The Breslow (partial) likelihood is

$$L_B(\theta) = \prod_{i=1}^k \frac{\exp(\theta' s_i)}{[\sum_{l \in \mathcal{R}_i} \exp(\theta' z_l)]^{d_i}}$$

In a stratified analysis, the partial likelihood is the product of the partial likelihood for each individual stratum. For example, in a study of the time to first infection from a surgery, the variables of a patient consist of TIME (time from surgery to the first infection), STATUS (an indicator of whether the observation time is censored, with value 2 identifying a censored time), Z1 and Z2 (explanatory variables thought to be related to the time to infection), and GRP (a variable identifying the stratum to which the observation belongs). The specification in PROC PHREG for fitting the Cox model using the Breslow likelihood is as follows:

```
proc phreg;
  model time*status(2)=z1 z2/ties=breslow;
  strata grp;
run;
```

To cast the likelihood of the conditional logit model in the form of the Breslow likelihood, consider m artificial observed times for each individual who chooses one of m alternatives. The k th alternative is chosen at time 1; the choices of all other alternatives (second choice, third choice, ...) are not observed and would have been chosen at some later time. So a choice variable is coded with an observed time value of 1 for the chosen alternative and a larger value, 2, for all nonchosen (unobserved or censored alternatives). For each individual, there is exactly one event time (1) and $m - 1$ nonevent times, and the risk set just prior to this event time consists of all the m alternatives. For individual j with alternative-specific characteristics Z_{jl} for the l th alternative to choose the k th alternative, the Breslow likelihood is then

$$L_B(\theta) = \frac{\exp(\theta' Z_{jk})}{\sum_{l=1}^m \exp(\theta' Z_{jl})}$$

This is precisely the probability that individual j chooses alternative k in a conditional logit model. By stratifying on individuals, you get the likelihood of the conditional logit model. Note that the observed time values of 1 and 2 are chosen for convenience; however, the censored times have to be larger than the event time to form the correct risk set.

Before you invoke PROC PHREG to fit the conditional logit, you must arrange your data in such a way that there is a survival time for each individual-alternative. In the example of travel demand, let SUBJECT identify the individuals, let TRAVTIME represent the travel time for each mode of transportation, and let CHOICE have a value 1 if the alternative is chosen and 2 otherwise. The CHOICE variable is used as the artificial time variable as well as a censoring variable in PROC PHREG. The following SAS statements reshape the data set TRAVEL into data set CHOICE and display the first nine observations:

```
data choice(keep=subject mode travtime choice);
  array times[3] autotime plantime trantime;
  array allmodes[3] $
    _temporary_ ('Auto' 'Plane' 'Transit');
  set travel;
  subject = _n_;
  do i = 1 to 3;
    mode = allmodes[i];
    travtime = times[i];
    choice = 2 - (chosen eq mode);
    output;
  end;
run;

proc print data=choice(obs=9);
run;
```

OBS	SUBJECT	MODE	TRAVTIME	CHOICE
1	1	Auto	10.0	2
2	1	Plane	4.5	1
3	1	Transit	10.5	2
4	2	Auto	5.5	1
5	2	Plane	4.0	2
6	2	Transit	7.5	2
7	3	Auto	4.5	2
8	3	Plane	6.0	2
9	3	Transit	5.5	1

Notice that each observation in TRAVEL corresponds to a block of three observations in CHOICE, exactly one of which is chosen.

The following SAS statements invoke PROC PHREG to fit the conditional logit model. The Breslow likelihood is requested by specifying TIES=BRESLOW. CHOICE is the artificial time variable, and a value of 2 identifies censored times. SUBJECT is used as a stratification variable.

```
proc phreg data=choice;
  model choice*choice(2)=travtime;
  strata subject;
  title 'Conditional Logit Model Using PHREG';
run;
```

Conditional Logit Model Using PHREG						
Analysis of Maximum Likelihood Estimates						
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square	Risk Ratio
TRAVTIME	1	-0.265495	0.10215	6.75511	0.0093	0.767

To study the relationship between the choice of transportation and the age of people making the choice, the analysis is based on the generalized logit model. You can use PROC CATMOD directly to fit the generalized logit model (refer to *SAS/STAT User's Guide, Vol. 1*). In the following invocation of PROC CATMOD, CHOSEN is the response variable and AGE is the explanatory variable:

```
proc catmod data=travel;
  direct age;
  model chosen=age;
  title 'Multinomial Logit Model Using Catmod';
run;
```

Generalized Logit Model Using CATMOD					
CATMOD PROCEDURE					
RESPONSE PROFILES					
Response CHOSEN					

	1	Auto			
	2	Plane			
	3	Transit			

ANALYSIS OF MAXIMUM-LIKELIHOOD ESTIMATES					
Effect	Parameter	Estimate	Standard Error	Chi-Square	Prob

INTERCEPT	1	3.0449	2.4268	1.57	0.2096
	2	2.7212	2.2929	1.41	0.2353
AGE	3	-0.0710	0.0652	1.19	0.2762
	4	-0.0500	0.0596	0.70	0.4013

Note that there are two intercept coefficients and two slope coefficients for AGE. The first INTERCEPT and the first AGE coefficients correspond to the effect on the probability of choosing auto over transit, and the second intercept and second age coefficients correspond to the effect of choosing plane over transit.

Let X_j be a $(p + 1)$ -vector representing the characteristics of individual j . The generalized logit model can be cast in the framework of a conditional model by defining the global parameter vector θ and the alternative-specific regressor variables Z_{jk} as follows:

$$\theta = \begin{bmatrix} \beta_1 \\ \beta_2 \\ \vdots \\ \beta_{m-1} \end{bmatrix}$$

$$Z_{j1} = \begin{bmatrix} X_j \\ 0 \\ \vdots \\ 0 \end{bmatrix}$$

$$Z_{j2} = \begin{bmatrix} 0 \\ X_j \\ 0 \\ \vdots \\ 0 \end{bmatrix}$$

$$\dots = \dots$$

$$\dots = \dots$$

$$Z_{j,m-1} = \begin{bmatrix} 0 \\ \vdots \\ 0 \\ X_j \end{bmatrix}$$

$$Z_{jm} = \begin{bmatrix} 0 \\ \vdots \\ 0 \end{bmatrix}$$

where the 0 is a $(p + 1)$ -vector of zeros. The probability that individual j chooses alternative k for the generalized logit model is put in the form that corresponds to a conditional logit model as follows:

$$\Pi_{jk} = \frac{\exp(\beta'_k X_j)}{\sum_{i=1}^m \exp(\beta'_i X_j)}$$

$$= \frac{\exp(\theta' Z_{jk})}{\sum_{i=1}^m \exp(\theta' Z_{ji})}$$

Here, the vector X_j representing the characteristics of individual j includes the element 1 for the intercept parameter (provided that the intercept parameters are to be included in the model).

By casting the generalized logit model into a conditional logit model, you can then use PROC PHREG to analyze the generalized logit model. In the example of travel demand, the alternative-specific variables AUTO, PLANE, AGEAUTO, and AGEPLANE are created from the individual characteristic variable AGE. The following SAS statements reshape the data set TRAVEL into data set CHOICE2 and display the first nine observations:

```
data choice2;
  array times[3] autotime plantime trantime;
  array allmodes[3] $
    _temporary_ ('Auto' 'Plane' 'Transit');
  set travel;
  subject = _n_;
  do i = 1 to 3;
    mode = allmodes[i];
    travtime = times[i];
    choice = 2 - (chosen eq mode);
  end;
```

```

auto = (i eq 1);
plane = (i eq 2);
ageauto = auto * age;
ageplane = plane * age;
output;
end;
keep subject mode travtime choice auto
plane ageauto ageplane;
run;

proc print data=choice2(obs=9);
run;

```

OBS	SUBJECT	MODE	TRAVTIME	CHOICE	AUTO	PLANE	AGEAUTO	AGEPLANE
1	1	Auto	10.0	2	1	0	32	0
2	1	Plane	4.5	1	0	1	0	32
3	1	Transit	10.5	2	0	0	0	0
4	2	Auto	5.5	1	1	0	13	0
5	2	Plane	4.0	2	0	1	0	13
6	2	Transit	7.5	2	0	0	0	0
7	3	Auto	4.5	2	1	0	41	0
8	3	Plane	6.0	2	0	1	0	41
9	3	Transit	5.5	1	0	0	0	0

The following SAS statements invoke PROC PHREG to fit the generalized logit model:

```

proc phreg data=choice2;
model choice*choice(2) = auto plane ageauto
ageplane;
strata subject;
title 'Generalized Logit Model Using PHREG';
run;

```

Generalized Logit Model Using PHREG The PHREG Procedure						
Analysis of Maximum Likelihood Estimates						
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square	Risk Ratio
AUTO	1	3.044945	2.42682	1.57429	0.2096	21.009
PLANE	1	2.721207	2.29289	1.40850	0.2353	15.199
AGEAUTO	1	-0.070967	0.06517	1.18592	0.2762	0.931
AGEPLANE	1	-0.050003	0.05958	0.70445	0.4013	0.951

By transforming individual characteristics into alternative-specific variables, the mixed logit model can be analyzed as a conditional logit model.

Analyzing the travel demand data for the effects of both travel time and age of individual requires the same data set as the generalized logit model, only now the TRAVTIME variable will be used as well. The following SAS statements use PROC PHREG to fit the mixed logit model:

```

proc phreg data=choice2;
model choice*choice(2)=auto plane ageauto
ageplane travtime;
strata subject;
title 'Mixed Logit Model Using PHREG';
run;

```

Mixed Logit Model Using PHREG The PHREG Procedure						
Analysis of Maximum Likelihood Estimates						
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square	Risk Ratio
AUTO	1	2.500694	2.39585	1.08943	0.2966	12.191
PLANE	1	-2.779213	3.52932	0.62010	0.4310	0.062
AGEAUTO	1	-0.078257	0.06332	1.52742	0.2165	0.925
AGEPLANE	1	0.016949	0.07439	0.05191	0.8198	1.017
TRAVTIME	1	-0.608466	0.27126	5.03147	0.0249	0.544

A special case of the mixed logit model is the conditional logit model with alternative-specific constants. Each alternative in the model can be represented by its own intercept, which captures the unmeasured desirability of the alternative.

```

proc phreg data=choice2;
model choice*choice(2)=auto plane travtime;
strata subject;
title 'Conditional Logit Model with '
'Alternative Specific Constants';
run;

```

Conditional Logit Model with Alternative-Specific Constants The PHREG Procedure						
Analysis of Maximum Likelihood Estimates						
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square	Risk Ratio
AUTO	1	-0.119661	0.70820	0.02855	0.8658	0.887
PLANE	1	-1.631449	1.24251	1.72405	0.1892	0.196
TRAVTIME	1	-0.486651	0.20725	5.51387	0.0189	0.615

With transit as the reference mode, the intercept for auto, which is negative, may reflect the inconvenience of having to drive over travelling by bus/train, and the intercept for plane may reflect the high expense of traveling by plane over bus/train.

Cross-Alternative Effects

Discrete choice models are often derived from the principle of maximum random utility. It is assumed that an unobserved utility V_k is associated with the k th alternative, and the response function Y is determined by

$$Y = k \Leftrightarrow V_k = \max\{V_l, 1 \leq l \leq m\}$$

Both the generalized logit and the conditional logit models are based on the assumption that V_1, \dots, V_m are independently distributed and each follows an extreme maxima value distribution (Hoffman and Duncan, 1988). An important property of such models is Independence from Irrelevant Alternatives (IIA); that is, the ratio of the choice probabilities for any two alternatives for a particular observation is not influenced systematically by any other alternatives. IIA can be

tested by fitting a model that contains all the cross-alternative effects and examining the significance of these effects. The cross-alternative effects pick up a variety of IIA violations and other sources of error in the model.

In the example of travel demand, there may be separate effects for the three travel modes and travel times. In addition, there may be cross-alternative effects for travel times. Not all the effects are estimable, only two of the three intercepts and three of the six cross-alternative effects can be estimated. The following SAS statements create the design variables for all the cross-alternative effects and display the first nine observations:

```

* Number of alternatives in each choice set;
%let m = 3;

data choice3;
  drop i j k autotime plantime trantime;

  * Values of the variable CHOSEN;
  array allmodes[&m] $
    _temporary_ ('Auto' 'Plane' 'Transit');

  * Travel times for the alternatives;
  array times[&m] autotime plantime trantime;

  * New variables that will contain the design;;
  array inters[&m]
    auto /*intercept for auto */
    plane /*intercept for plane */
    transit; /*intercept for transit */
  array cross[&eval(&m * &m)]
    timeauto /*time of auto alternative */
    planauto /*cross effect of plane on auto */
    tranauto /*cross effect of transit on auto */
    autoplan /*cross effect of auto on plane */
    timeplan /*time of plane alternative */
    tranplan /*cross effect of transit on plane*/
    autotran /*cross effect of auto on transit */
    plantran /*cross effect of plane on transit*/
    timetran; /*time of transit alternative */
  set travel;

  subject = _n_;

  * Create &m observations for each choice set;
  do i = 1 to &m;
    mode = allmodes[i]; /* this alternative */
    travtime = times[i]; /* travel time */
    choice = 2 - (chosen eq mode); /* 1 - chosen */
    do j = 1 to &m;
      inters[j] = (i eq j); /* mode indicator */
      do k = 1 to &m;
        * (j=k) - time, otherwise, cross effect;
        cross[&m*(j-1)+k]=times[k]*inters[j];
      end;
    end;
  output;
end;
run;

proc print data=choice3(obs=9);
  var subject mode travtime choice
    auto plane transit
    timeauto timeplan timetran

```

```

autoplan autotran planauto
plantran tranauto tranplan;
run;

```

	S	T	T	T	T	A	A	P	P	T	T				
	U	R	I	I	I	U	U	L	L	R	R				
	B	A	C	R	M	M	N	T	T	A	A				
	J	V	H	P	A	E	E	O	O	N	N				
	M	T	O	A	L	A	P	T	P	T	A				
	O	I	I	U	A	S	U	L	R	L	R				
	B	C	D	M	C	T	N	I	T	A	A				
	S	T	E	E	R	O	B	T	O	N	N				
1 1	Auto	10.0	2	1	0	0	10.0	0.0	0.0	0.0	0.0	4.5	0.0	10.5	0.0
2 1	Plane	4.5	1	0	1	0	0.0	4.5	0.0	10.0	0.0	0.0	0.0	0.0	10.5
3 1	Transit	10.5	2	0	0	1	0.0	0.0	10.5	0.0	10.0	0.0	0.0	4.5	0.0
4 2	Auto	5.5	1	1	0	0	5.5	0.0	0.0	0.0	0.0	4.0	0.0	7.5	0.0
5 2	Plane	4.0	2	0	1	0	0.0	4.0	0.0	5.5	0.0	0.0	0.0	0.0	7.5
6 2	Transit	7.5	2	0	0	1	0.0	0.0	7.5	0.0	5.5	0.0	4.0	0.0	0.0
7 3	Auto	4.5	2	1	0	0	4.5	0.0	0.0	0.0	0.0	6.0	0.0	5.5	0.0
8 3	Plane	6.0	2	0	1	0	0.0	6.0	0.0	4.5	0.0	0.0	0.0	0.0	5.5
9 3	Transit	5.5	1	0	0	1	0.0	0.0	5.5	0.0	4.5	0.0	6.0	0.0	0.0

PROC PHREG allows you to specify TEST statements for testing linear hypotheses of the parameters. The test is a Wald test, which is based on the asymptotic normality of the parameter estimators. The following SAS statements invoke PROC PHREG to fit the so called "Mother Logit" model that includes all the cross-alternative effects. The TEST statement, with label IIA, specifies the null hypothesis that cross-alternative effects AUTOPLAN, PLANTRAN, and TRANAUTO are 0. Since only three cross-alternative effects are estimable and these are the first cross-alternative effects specified in the model, they account for all the cross-alternative effects in the model.

```

proc phreg data=choice3;
  model choice*choice(2)=auto plane transit
    timeauto timeplan timetran
    autoplan plantran tranauto
    planauto tranplan autotran;
  IIA: test autoplan,plantran,tranauto;
  strata subject;
  title 'Mother Logit Model';
run;

```

Mother Logit Model						
Testing Global Null Hypothesis: BETA=0						
Criterion	Without Covariates	With Covariates	Model Chi-Square			
-2 LOG L	46.142	24.781	21.361 with 8 DF	(p=0.0062)		
Score			15.406 with 8 DF	(p=0.0517)		
Wald			6.240 with 8 DF	(p=0.6203)		
Analysis of Maximum Likelihood Estimates						
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square	Risk Ratio
AUTO	1	-0.738126	3.05934	0.05821	0.8093	0.478
PLANE	1	-3.624346	3.48049	1.08437	0.2977	0.027
TRANSIT	0	0				
TIMEAUTO	1	-2.234372	1.89924	1.38404	0.2394	0.107
TIMEPLAN	1	-0.101119	0.68622	0.02171	0.8828	0.904
TIMEPLAN	1	0.097849	0.70096	0.01949	0.8890	1.103
AUTOPLAN	1	0.444953	0.68616	0.42051	0.5167	1.560
PLANTRAN	1	-0.532338	0.63481	0.70321	0.4017	0.587
TRANAUTO	1	1.662977	1.51196	1.20974	0.2714	5.275
PLANAUTO	0	0				
TRANPLAN	0	0				
AUTOTRAN	0	0				

Mother Logit Model			
Linear Hypotheses Testing			
Label	Wald Chi-Square	DF	Pr > Chi-Square
IIA	1.6526	3	0.6475

The χ^2 statistic for the Wald test is 1.6525 with 3 degrees of freedom, indicating that the cross-alternative effects are not statistically significant ($p = .6475$). A generally more preferable way of testing the significance of the cross-alternative effects is to compare the likelihood of the "Mother logit" model with the likelihood of the reduced model with the cross-alternative effects removed. The following SAS statements invoke PROC PHREG to fit the reduced model:

```
proc phreg data=choice3;
  model choice*choice(2)=auto plane transit
    timeauto timeplan timetran;
  strata subject;
  title 'Reduced Model without '
    'Cross-Alternative Effects';
run;
```

Reduced Model without Cross-Alternative Effects						
The PHREG Procedure						
Testing Global Null Hypothesis: BETA=0						
Criterion	Without Covariates	With Covariates	Model	Chi-Square		
-2 LOG L	46.142	27.153	18.989 with 5 DF	(p=0.0019)		
Score	.	.	14.460 with 5 DF	(p=0.0129)		
Wald	.	.	7.342 with 5 DF	(p=0.1964)		
The PHREG Procedure						
Analysis of Maximum Likelihood Estimates						
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square	Risk Ratio
AUTO	1	1.715783	1.80467	0.90392	0.3417	5.561
PLANE	1	-3.600732	3.30555	1.18657	0.2760	0.027
TRANSIT	0	0
TIMEAUTO	1	-0.795432	0.36327	4.79462	0.0285	0.451
TIMEPLAN	1	0.121619	0.58954	0.04256	0.8366	1.129
TIMETRAN	1	-0.421843	0.25733	2.68729	0.1012	0.656

The chi-squared statistic for the likelihood ratio test of IIA is $(27.153 - 24.781) = 2.372$, which is not statistically significant ($p = .4989$) when compared to a χ^2 distribution with 3 degrees of freedom. This is consistent with the previous result of the Wald test.

Final Comments

For some discrete choice problems, the number of available alternatives is not the same for each individual. For example, in a study of consumer brand choices of laundry detergents as prices change, data are pooled from different locations, not all of which

offer a brand that contains potash. The varying choice sets across individuals can easily be accommodated in PROC PHREG. For individual j who chooses from a set of m_j alternatives, consider m_j artificial times in which the chosen alternative has an event time 1 and the nonchosen alternatives have a censored time of 2. The analysis is carried out in the same fashion as illustrated in the previous section.

Unlike the example of travel demand in which data for each individual are provided, choice data are often given in aggregate form, with choice frequencies indicating the repetition of each choice. One way of dealing with aggregated data is to expand the data to the individual level and carry out the analysis as if you have nonaggregated data. This approach is generally not recommended because it defeats the purpose of having a smaller aggregated data set. PROC PHREG provides a FREQ statement that allows you to specify a variable that identifies the frequency of occurrence of each observation. However, with the specification of a FREQ variable, the artificial event time is no longer the only event time in a given stratum, but has ties of the given frequency. With proper stratification, the Breslow likelihood is proportional to the likelihood of the conditional logit model. Thus PROC PHREG can be used to obtain parameter estimates and hypothesis testing results for the choice models. Technical Support note TS-237* contains an example of aggregated choice data analyzed by using PROC PHREG.

The TIES=DISCRETE option should not be used instead of the TIES=BRESLOW option. This is especially detrimental with aggregated choice data because the likelihood that PROC PHREG is maximizing may no longer be the same as the likelihood of the conditional logit model. TIES=DISCRETE corresponds to the discrete logistic model for genuinely discrete time scale, which is also suitable for the analysis of case-control studies when there is more than one case in a matched set (Gail, Lubin, and Rubinstein, 1981). For nonaggregated choice data, all TIES= options give the same results; however, the resources required for the computation are not the same, with TIES=BRESLOW being the most efficient.

Once you have a basic understanding of how PROC PHREG works, you can use it to fit a variety of models

*A postscript file (techsup/download/technote/ts273.ps) is available by anonymous ftp from ftp.sas.com or via the World Wide Web at

<http://www.sas.com/techsup/download/technote/>.

for the discrete choice data. The major involvement in such a task lies in reorganizing the data to create the observations necessary to form the correct risk sets and the appropriate design variables. There are many options in PROC PHREG that can also be useful in the analysis of discrete choice data. For example, the OFFSET= option allows you to restrict the coefficient of an explanatory variable to the value of 1; the SELECTION= option allows you to specify one of four methods for selecting variables into the model; the OUTEST= option allows you to specify the name of the SAS data set that contains the parameter estimates, based on which you can easily compute the predicted probabilities of the alternatives.

This article deals with estimating parameters of discrete choice models. There is active research in the field of marketing research to use design of experiments to study consumer choice behavior. If you are interested in this area, refer to Carson et al. (1994), Kuhfeld et al. (1994), and Lazari et al. (1994).

REFERENCES

- Agresti, A. (1990) *Categorical Data Analysis*. New York: John Wiley & Sons.
- Breslow, N. and Day, N.E. (1980), *Statistical Methods in Cancer Research, Vol. II: The Design and Analysis of Cohort Studies*, Lyon: IARC.
- Carson, R.T.; Louviere, J.J.; Anderson, D.A.; Arabie, P.; Bunch, D.; Hensher, D.A.; Johnson, R.M.; Kuhfeld, W.F.; Steinberg, D.; Swait, J.; Timmermans, H.; and Wiley, J.B. (1994). "Experimental Analysis of Choice," *Marketing Letters*, 5(4), 351-368.
- Gail, M.H., Lubin, J.H., and Rubinstein, L.V. (1981), "Likelihood calculations for matched case-control studies and survival studies with tied death times," *Biometrika*, 68, 703-707.
- Hoffman, S.D. and Duncan, G.J. (1988), "Multinomial and conditional logit discrete-choice models in Demography," *Demography*, 25 (3), 415-427.
- Kuhfeld, W.F., Tobias, R.D., and Garratt, M. (1994), "Efficient Experimental Design with Marketing Research Applications," *Journal of Marketing Research*, 31, 545-557.
- Lazari, A.G. and Anderson, D.A. (1994), "Designs of Discrete Choice Set Experiments for Estimating Both Attribute and Availability Cross Effects," *Journal of Marketing Research*, 31, 375-383.
- McFadden, D. (1973), "Conditional logit analysis of qualitative choice behavior," in P. Zarembka (Ed.) *Frontiers in Econometrics*, New York: Academic

Press, Inc.

McKelvey, R.D. and Zavoina, W. (1975), "A statistical model for the analysis of ordinal level dependent variables," *Journal of Mathematical Sociology*, 4, 103-120.

SAS Institute Inc. (1989), *SAS/STAT User's Guide, Vol. 1, Version 6, Fourth Edition*, Cary: NC: SAS Institute Inc.

SAS Institute Inc. (1992), SAS Technical Report P-229. *SAS/STAT Software: Changes and Enhancements, Release 6.07*, Cary, NC: SAS Institute Inc.

SAS Institute Inc. (1993), "Multinomial logit, discrete choice modeling", Technical Support Note TS-273, Cary, NC: SAS Institute Inc.

SAS and SAS/STAT are trademarks or registered trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.